

**Les outils à maîtriser avant de commencer :****B. Les références relatives, absolues et mixtes**

Le point fort d'Excel est de nous éviter la saisie répétitive d'une même formule, grâce à l'utilisation d'une recopie "intelligente". En effet, le simple fait de recopier une formule dans la verticale (changement de lignes) ou dans l'horizontale (changement de colonnes) vers une autre cellule entraîne la modification des références utilisées dans le calcul : elles s'adaptent automatiquement.

Pour une meilleure compréhension, nous vous proposons de reproduire les différents exemples qui vont illustrer nos propos.

**1) Références relatives**

Nous parlons de références relatives lorsque suite à la recopie d'une formule vers d'autres cellules, les références qui la composent changent automatiquement pour s'adapter au lieu où elles sont copiées.

Voici un exemple d'utilisation :

	A	B	C	D	E
1	<b>Produits</b>	<b>Quantités</b>	<b>Prix unitaire</b>	<b>Total</b>	
2	pommes	10	2.00	=B2*C2	

Nous avons mis en place en D2 la formule de calcul pour obtenir le total. Nous allons la valider et la recopier de D3 à D5 en utilisant la poignée de recopie.



*Vous pouvez également utiliser le copier-coller ou l'outil Remplissage vers le bas du groupe Édition de l'onglet Accueil.*

	A	B	C	D	E
1	<b>Produits</b>	<b>Quantités</b>	<b>Prix unitaire</b>	<b>Total</b>	
2	pommes	10	2.00	20.00	D2=B2*C2
3	pêches	15	3.00	45.00	D3=B3*C3
4	poires	20	2.20	44.00	D4=B4*C4
5	abricots	5	4.00	20.00	D5=B5*C5
6					

La formule est recopiée et les références sont automatiquement modifiées afin qu'elles puissent correspondre aux lignes.

Les références sont dites relatives.

Cependant, cette technique peut devenir gênante dans certains cas.

## 2. Références absolues

Nous avons besoin parfois d'effectuer un calcul composé d'éléments qui doivent être relatifs et d'éléments qui doivent être fixes, c'est-à-dire identiques pour chaque cellule.

C'est le cas de l'exemple suivant. Il fait appel à une donnée (le prix unitaire) qui doit rester identique à tous nos calculs :

	A	B	C	D	E
1	Prix unitaire	25			
2					
3		usine1	usine2	usine3	
4	quantités produites	1500	1000	2000	
5	CA prévisionnel	=B4*B1			
6					

Or, si nous utilisons la poignée de recopie, nous obtenons le résultat suivant :

	A	B	C	D	E
1	Prix unitaire	25			
2					
3		usine1	usine2	usine3	
4	quantités produites	1500	1000	2000	
5	CA prévisionnel	37500	0	0	
6		B5=B4*B1	C5=C4*C1	D5=D4*D1	
7					

Pour faire en sorte que le prix unitaire (cellule B1) soit utilisé dans la formule de chaque calcul, il faut le désigner par une référence dite absolue, en utilisant le caractère \$.

La référence B1 doit s'écrire \$B\$1. Ce caractère, placé avant la colonne, interdit lors de la recopie horizontale le changement de colonne, et placé devant le numéro de la ligne, il interdit lors de la recopie verticale le changement de ligne. Ce qui, au final, nous donne ceci :

	A	B	C	D	E
1	Prix unitaire	25			
2					
3		usine1	usine2	usine3	
4	quantités produites	1500	1000	2000	
5	CA prévisionnel	37500	25000	50000	
6		B5=B4*\$B\$1	C5=C4*\$B\$1	D5=D4*\$B\$1	
7					



### 3. Références mixtes

Nous avons également des cas où les références relatives ou absolues ne sont pas adaptées, comme dans l'exemple ci-dessous.

	A	B	C	D	E
1		usine1	usine2	usine3	
2	Coef prévisionnel	1.1	1.3	1.2	
3					
4	production de base	usine1	usine2	usine3	
5	1000	=A5*B2			
6	1500				
7	1600				
8					

Chaque usine doit utiliser les productions de base (colonne A), tout en utilisant le coefficient d'évolution qui lui est propre.

Chaque usine doit utiliser les productions de base (colonne A), tout en utilisant le coefficient d'évolution qui lui est propre.

Avec des références relatives, nous aurions :

	A	B	C	D	E	F	G
1		usine1	usine2	usine3			
2	Coef prévisionnel	1.1	1.3	1.2			
3							
4	production de base	usine1	usine2	usine3			
5	1000	1100	1430	1716	B5=A5*B2	C5=B5*C2	D5=C5*D2
6	1500	0	0	0	B6=A6*B3	C6=B6*C3	D6=C6*D3
7	1600	#VALEUR!	#VALEUR!	#VALEUR!	B7=A7*B4	C7=B7*C4	D7=C7*D4
8							

Nous constatons qu'en B6 la référence A6 est correcte mais pas la référence B3 qui aurait dû rester sur le coef en B2, tandis qu'en C6 plus rien n'est correct.

Lorsque la recopie des données se fait vers la droite, la colonne A (A5, A6 et A7) doit se retrouver dans toutes nos formules. Quand elle se fait vers le bas, c'est le numéro de la ligne qui doit se retrouver partout (B2, C2, D2).

C'est là qu'entrent en jeu les références mixtes qui peuvent se présenter ainsi :

**\$A5** : signifie que si la recopie se fait à l'horizontale, la colonne A est absolue, donc fixe, alors que si la recopie se fait à la verticale, le 5 est relatif, il changera selon la ligne où se trouve la formule.

**A\$5** : signifie que si la recopie se fait à l'horizontale, la colonne A est relative, elle changera selon la colonne où se trouve la formule ; si la recopie se fait à la verticale, le 5 est absolu, donc fixe.

En résumé, pour notre exemple la bonne formule est  $=\$A5*B\$2$ , ce qui donne :

	A	B	C	D	E	F	G
1		<b>usine1</b>	<b>usine2</b>	<b>usine3</b>			
2	<b>Coef prévisionnel</b>	<b>1.1</b>	<b>1.3</b>	<b>1.2</b>			
3							
4	<b>production de base</b>	<b>usine1</b>	<b>usine2</b>	<b>usine3</b>			
5	1000	1100	1300	1200	$B5=\$A5*B\$2$	$C5=\$A5*C\$2$	$D5=\$A5*D\$2$
6	1500	1650	1950	1800	$B6=\$A6*B\$2$	$C6=\$A6*C\$2$	$D6=\$A6*D\$2$
7	1600	1760	2080	1920	$B7=\$A7*B\$2$	$C7=\$A7*C\$2$	$D7=\$A7*D\$2$
8							

### C Nommer une cellule

Il n'est pas toujours nécessaire de bloquer une référence en utilisant le caractère \$, il est aussi possible de nommer les cellules.

Le fait de nommer une cellule est particulièrement intéressant lorsque les formules sont complexes. Mais cela peut aussi avoir un intérêt dans des calculs simples par le fait qu'un nom est souvent plus parlant qu'une référence.

#### 1. Créer un nom de manière rapide

Nous allons le faire à partir d'un exemple :

	A	B	C
1	<b>Taux TVA</b>	<b>19.60%</b>	
2			
3	<b>HT</b>	<b>Montant TVA</b>	
4	<b>1500</b>		
5	<b>2100</b>		
6	<b>3000</b>		
7			

☞ Pour nommer la cellule B1 qui contient le taux de TVA, cliquez dans la zone Nom de la barre de formule puis saisissez TVA.

	A	B	C
1	Taux TVA	19.60%	
2			
3	HT	Montant TVA	
4	1500		
5	2100		
6	3000		

☞ Validez par .

☞ Cliquez en B4 puis saisissez `=A4*TVA`.

	A	B	C
1	Taux TVA	19.60%	
2			
3	HT	Montant TVA	
4	1500	=A4*TVA	
5	2100		
6	3000		
7			

La cellule B1 en se nommant TVA devient absolue. En recopiant votre formule vers le bas, le nom TVA sera répété à chaque ligne.

## 2. Gestionnaire de noms

Vous voulez modifier ou supprimer un nom, vous passerez par le gestionnaire de noms :  
Onglet Formules - groupe Noms définis - bouton Gestionnaire de noms

Vous trouverez dans la boîte de dialogue tous les noms créés dans le classeur avec des informations telles que les références, la valeur, etc.

Dès que vous sélectionnez un nom, les boutons Supprimer et Modifier s'activent.



*Les noms ne doivent contenir ni espaces, ni accents. Évitez les noms du type AB3 ou TVA1 car ce sont des références de cellules existantes et ne seront donc pas acceptées.*



## D. La fonction SI

### 1. Une question, deux réponses possibles

Fonction incontournable qui rend bien des services. Elle permet de choisir la réponse à obtenir selon une condition.

Sa structure est la suivante :

**=SI**(question ; valeur si réponse positive ; valeur si réponse négative)

Elle pourrait se traduire par :

SI question vérifiée ALORS renvoyer valeur SINON renvoyer autre valeur.

**Question** : la question consiste à comparer le contenu d'une cellule à un élément en utilisant les signes égal =, supérieur >, inférieur <, supérieur ou égal >=, inférieur ou égal <=, différent <>. Cette question est de type logique, c'est-à-dire entraînant la réponse oui ou non (vrai ou faux).

Cet élément peut être une valeur numérique ( $A1 > 1000$ ), une valeur texte à écrire entre guillemets ( $A1 = \text{"bonjour"}$ ), une référence de cellule ( $A1 = B1$ ), une formule ( $A1 > \text{somme}(B1:C1)$ ).

**Valeur si réponse positive** : correspond à la valeur à renvoyer si la réponse à la question est positive (vrai). Cette valeur peut être numérique (100), texte ("bravo") ou le résultat d'un calcul ( $A1 * 10\%$ ).

**Valeur si réponse négative** : correspond à la valeur à renvoyer si la réponse à la question est négative (faux) ; elle peut être numérique, texte ou le résultat d'un calcul.

Voici un exemple simple :

	A	B	C
1	temps	action	
2	soleil	=SI(A2="pluie";"parapluie";"parasol")	
3	pluie		
4			

À traduire par : SI en A2 est inscrite la valeur textuelle pluie, ALORS renvoyer le texte parapluie SINON renvoyer le texte parasol.

Résultat après recopie de la formule en B3 :

	A	B	C
1	temps	action	
2	soleil	parasol	
3	pluie	parapluie	
4			

## 2. Une question, plusieurs réponses possibles

Deux réponses c'est déjà bien, mais c'est parfois insuffisant. Nous pouvons avoir besoin de 3, 4 réponses ou plus. Nous pouvons aller jusqu'à 65 réponses depuis la version 2007 d'Excel.

Dès que le stade de deux réponses est dépassé, une nouvelle fonction SI imbriquée dans la première est nécessaire.

Voici la syntaxe pour obtenir 3 réponses :

`=SI(question ; valeur si réponse positive ; SI(question ; valeur si réponse positive ; valeur si réponse négative))`

ou

`=SI(question ; SI(question ; valeur si réponse positive ; valeur si réponse négative) ; valeur si réponse négative)`

Dans le premier cas, la nouvelle fonction SI a pris la place de la réponse négative de la première fonction SI ; dans le deuxième cas, elle a pris la place de la réponse positive de la première fonction SI.

Essayons cette fonction dans un exemple concret : nous avons besoin de déterminer un prix de revient selon la composition d'un article.

Si un vêtement est fabriqué en coton, le prix de base sera multiplié par 1.5, s'il est fabriqué en lin le prix sera multiplié par 2, et par 3 s'il est fabriqué en soie. Cela donnera ceci :

	A	B	C
1	<b>Prix base</b>	<b>10</b>	
2			
3	<b>matière</b>	<b>prix du vêtement</b>	
4	coton	<code>=SI(A4="coton";\$B\$1*1.5;SI(A4="soie";\$B\$1*3;\$B\$1*2))</code>	
5	lin		
6	soie		
7			

La cellule B1 a été bloquée par les caractères \$ pour permettre la recopie de la formule vers le bas.

Nous obtenons au final :

	A	B
1	<b>Prix base</b>	<b>10</b>
2		
3	<b>matière</b>	<b>prix du vêtement</b>
4	coton	15
5	lin	20
6	soie	30



Vous avez dû remarquer que seules deux matières ont été testées, la troisième est déduite implicitement dans le sens que si ce n'est pas coton ni soie c'est forcément la seule qui reste c'est-à-dire le lin.

Cette formule aurait pu se présenter de différentes manières selon le choix des questions.

=SI(A4="lin";\$B\$1\*2;SI(A4="soie";\$B\$1\*3;\$B\$1\*1.5)) ou

=SI(A4="soie";\$B\$1\*3;SI(A4="coton";\$B\$1\*1.5;\$B\$1\*2)) ou

=SI(A4="coton";\$B\$1\*1.5;SI(A4="lin";\$B\$1\*2;\$B\$1\*3)) ou etc.

À noter le positionnement des réponses dans chaque formulation.

### 3. Plusieurs questions dans une fonction SI

Nous pouvons être amenés parfois à poser plusieurs questions pour obtenir ne serait-ce que deux réponses.

Ces questions peuvent porter sur la même cellule : si la cellule A1 est supérieure à 10 et inférieure à 100, alors...

Ou sur des cellules différentes : si la cellule A1 est égale à "bonjour" ou A2 égale à "salut", alors... etc.

Nous aurons besoin de deux nouvelles fonctions : ET et OU.

#### a) Fonction ET

Cette fonction utilisée seule renvoie la valeur VRAI si toutes les conditions (questions) sont vérifiées, et la valeur FAUX si au moins l'une des conditions n'est pas vérifiée.

Sa syntaxe est la suivante :

=ET(question1;question2;...;question...)



Depuis la version 2007, une fonction peut contenir jusqu'à 255 questions (30 auparavant).

#### Exemple

Si le prix d'un produit est compris entre 10 et 20, l'augmenter de 10 %, sinon le laisser tel quel :

	A	B	C	D
1	<b>produit</b>	<b>prix</b>	<b>nouveau prix</b>	
2	produit1		=SI(ET(B2>10;B2<20);B2*(1+10%);B2)	
3	produit2	11		
4	produit3	19		



	A	B	C
1	<b>produit</b>	<b>prix</b>	<b>nouveau prix</b>
2	produit1	5	5
3	produit2	11	12.1
4	produit3	19	20.9
5			

### b) Fonction OU

Comme la fonction ET, cette fonction utilisée seule renvoie la valeur VRAI ou FAUX.

Sa syntaxe est la suivante :

=OU(question1;question2;...;question255)

#### Exemple

Si le jeu 1 ou le jeu 2 est gagné alors afficher Sélectionné sinon afficher Éliminé.

	A	B	C	D	E
1		<b>jeu1</b>	<b>jeu2</b>	<b>Résultat</b>	
2	participant1	gagné	perdu	=SI(OU(B2="gagné";C2="gagné");"Sélectionné";"Éliminé")	
3	participant2	gagné	gagné		
4	participant3	perdu	perdu		
5	participant4	perdu	gagné		
6					

	A	B	C	D
1		<b>jeu1</b>	<b>jeu2</b>	<b>Résultat</b>
2	participant1	gagné	perdu	Sélectionné
3	participant2	gagné	gagné	Sélectionné
4	participant3	perdu	perdu	Éliminé
5	participant4	perdu	gagné	Sélectionné